Design Project

06-100 Fall 2014

Mathematical Optimization



Bruno Abreu Calfa

The Problem



- Decide recycle fraction given alternatives
  - Separation
  - Reactor operating temperature
- Maximize profit of operation

The Solution Approach

Decision Making Alternatives Objective → Mathematical Optimization

- Basic steps
  1. Model the problem mathematically
  2. Choose appropriate solver for the type of model
- In general

Maximize/Minimize Objective Function ← criterion
subject to Constraints ← available resources

direction (best) → Maximize/Minimize

Simple Two-Variable Example

$$\min_{x,y} \quad f(x,y) = x^2 + y^2$$

$$\text{s.t.} \quad h_1(x,y) = \frac{3}{2}x - y = 1$$

$$h_2(x,y) = \frac{2}{3}x + y = 4$$





Design Problem Formulation #1

- **Objective**
  - Maximize profit
- **Decision Variables**
  - Conversion (X), Selectivity (S), Recycle Fraction (R), Formaldehyde Recovery Fraction (F_F)
- **"State" Variables (can be calculated from decision variables)**
  - Molar flow rates of methane in stream 1 (n_{M1}), formaldehyde in stream 7 (n_{F7}), water in stream 7 (n_{W7}), stream 3 (n_3)
- **Constraints**
  - Mole balances
  - Specification on formaldehyde/water mixture

Design Problem Formulation #1

- **Objective Function**

$$\text{Profit} = 3600 \cdot (\text{F}_{\text{val}} \cdot n_{F7} - \text{M}_{\text{cost}} \cdot n_{M1} - \text{React}_{\text{cost}} - \text{Sep}_{\text{cost}} \cdot n_3)$$

where F_{val} is the value of formaldehyde (e.g., \$16.52/kmol), M_{cost} is the cost of methane (e.g., \$0.969/kmol), $React_{cost}$ is the operating cost of the reactor at a given temperature (e.g., \$0.139/s), and Sep_{cost} is the separator cost (e.g., \$0.000267/kmol).

  – Note that the objective function is linear in the variables n_{F7}, n_{M1}, n_3.

Design Problem Formulation #1

- **Constraints**

$$n_{M1} = 9 - 9R + 9XSR + 9X(1-S)R$$

$$n_{F7} = F_F \left[9XS + \frac{9XSR(1-F_F)}{1-R+F_F R} \right]$$

$$n_{W7} = 9X(2-S)$$

$$n_3 = 10 + \frac{9XSR(1-F_F)}{1-R+F_F R} + \left(\frac{1}{1-R}\right) 9X(1-S) + \left(\frac{0.79}{0.21}\right)\left(\frac{1}{1-R}\right)[1 - R + 9XSR + 18XR(1-S)]$$

$$\frac{n_{F7}}{n_{F7} + n_{W7}} \geq 0.37$$

Algebraic Modeling System (AMS)

- Software environment that supports modeling of algebraic optimization models and offers links to specialized solvers.
- Some examples:
  - GAMS, AIMMS, AMPL, Pyomo.
- Main idea:



GAMS

- **G**eneral **A**lgebraic **M**odeling **S**ystem
  - http://www.gams.com/
- Model is written in a text file with extension .gms
- Solution is written to the corresponding listing file .lst
- Can use the GAMS IDE (Integrated Development Environment) to write models and view solution.
- Download GAMS for specific platform (Windows, Linux, Mac)
  - http://www.gams.com/download/
- Demo version: solves small problems
  - Valid license required to solve large problems
  - Some solvers require specific license

Writing Model in GAMS (I/III)

```
* Problem parameters
PARAMETERS
        M_cost          Cost of methane in $ per kmol /0.969/
        F_Value         Value of formaldehyde in $ per kmol /16.52/
        React_Cost      Cost for the reactor in $ per s /0.139/
        Sep_Cost        Separator cost in $ per kmol /0.000267/
;

* Decision variables
POSITIVE VARIABLES
        X       Conversion
        S       Selectivity
        R       Recycle fraction
        Ff      Formaldehyde recovery fraction
;

* Bounds on decision variables
X.up = 1;
S.up = 1;
R.up = 1;
Ff.up = 1;

* Objective function variable
VARIABLE
        Profit  Total profit of the system
;

* "State" variables
POSITIVE VARIABLES
        n_M1    Molar flow rate of M in stream 1
        n_F7    Molar flow rate of F in stream 7
        n_W7    Molar flow rate of W in stream 7
        n_3     Molar flow rate of stream 3
;
```

Defining model parameters (costs, values)

Defining nonnegative *decision* variables and setting their upper bounds

Defining objective function variable (profit)

Defining nonnegative *state* variables (flows)

Writing Model in GAMS (II/III)

```
X.fx = 0.008;
S.fx = 0.74;
Ff.fx = 0.5;

* Define constraints
EQUATIONS
         balM1   Mole balance for M in stream 1
         balF7   Mole balance for F in stream 7
         balW7   Mole balance for W in stream 7
         bal3    Overall mole balance for stream 3
         SpecF   Specification for F composition
         ObjFcn  Objective function (total cost calculation)
;

balM1..
n_M1 =e= 9 - 9*R + 9*X*S*R + 9*X*(1 - S)*R;

balF7..
n_F7*(1 - R + Ff*R) =e= Ff*(9*X*S*(1 - R + Ff*R) + 9*X*S*R*(1 - Ff));

balW7..
n_W7 =e= 9*X*(2 - S);

bal3..
n_3*(1 - R + Ff*R)*(1 - R) =e= 10*(1 - R + Ff*R)*(1 - R) + 9*X*S*R*(1 - Ff)*(1 - R) +
                               9*X*(1 - S)*(1 - R + Ff*R) +
                               0.79/0.21*(1 - R + 9*X*S*R + 18*X*R*(1 - S))*(1 - R + Ff*R);

SpecF..
n_F7 =g= 0*0.37*(n_F7 + n_W7);

ObjFcn..
Profit =e= 3600*(F_Value*n_F7 - M_Cost*n_M1 - React_Cost - Sep_Cost*n_3);
```

Fixing some of the decision variables to values

Defining constraints: mole balances, specification on F/W ratio, objective function

Carnegie Mellon ENGINEERING

Writing Model in GAMS (III/III)

```
* Declare optimization model (include all equations)
MODEL IntroDesign /all/;

* Specify solver and other options
OPTIONS
        NLP = BARON,
        optcr = 0,
        decimals = 6,
        limrow = 0,
        limcol = 0
;
$OFFSYMXREF
$OFFSYMLIST

* Solve problem
SOLVE IntroDesign USING NLP MAXIMIZING Profit;

* Post-processing paramaters
PARAMETERS
        Opt_F_Recov
        Opt_F_Value
        Opt_M_Cost
        Opt_Sep_Cost
;

Opt_F_Recov = n_F7.l/(n_F7.l + n_W7.l);
Opt_F_Value = F_Value*n_F7.l;
Opt_M_Cost = M_Cost*n_M1.l;
Opt_Sep_Cost = Sep_Cost*n_3.l;

* Display solution
DISPLAY Profit.l, R.l, X.l, S.l, n_M1.l, n_F7.l, n_W7.l, n_3.l;
DISPLAY Opt_F_Recov, Opt_F_Value, Opt_M_Cost, Opt_Sep_Cost;
```

- Defining model to include *all* constraints defined
- Setting options for solver and display of results
- Solving model
- Defining post-processing parameters and displaying results

Notes

- Some equations were written in GAMS differently from the ones on slide 6. The mole balance for F in stream 7 (constraint `bal7`) is written in GAMS as follows:

$$n_{F7}(1 - R + F_F R) = F_F \left[9XS(1 - R + F_F R) + 9XSR(1 - F_F) \right]$$

Note that there are no variables in a denominator as in the original equation. It is a good practice to avoid fractions involving variables when solving problems numerically (avoids, sometimes unnecessarily, division by zero). The mole balance for stream 3 (constraint `bal3`) was also written in GAMs without denominators.

- This model is nonlinear, since it contains product of variables. It is named a Nonlinear Optimization or Nonlinear Programming (NLP) problem.
- Nonlinear problems may have multiple solutions. The solver BARON guarantees to find the *global* best solution, if one exists.

Results

- Results <u>without</u> the 37% specification of F/W (constraint `SpecF`):

Profit [$]	2,186.26
R	0.996812
X	0.008000
S	0.740000
n_{M1} [kmol/s]	0.100464
n_{F7} [kmol/s]	0.053111
n_{W7} [kmol/s]	0.090720
n_3 [kmol/s]	126.386911

- If `SpecF` constraint is considered, R gets very close to 1 and the solution becomes numerically sensitive (essentially, division by zero problem).